<commit_before>// Install CoreOS to disk
resource "matchbox_group" "coreos-install" {
  count   = "${length(var.tectonic_metal_controller_names) + length(var.tectonic_metal_worker_names)}"
  name    = "${format("coreos-install-%s", element(concat(var.tectonic_metal_controller_names, var.tectonic_metal_worker_names), count.index))}"
  profile = "${matchbox_profile.coreos-install.name}"

  selector {
    mac = "${element(concat(var.tectonic_metal_controller_macs, var.tectonic_metal_worker_macs), count.index)}"
  }

  metadata {
    coreos_channel     = "${var.tectonic_cl_channel}"
    coreos_version     = "${var.tectonic_metal_cl_version}"
    ignition_endpoint  = "${var.tectonic_metal_matchbox_http_endpoint}/ignition"
    baseurl            = "${var.tectonic_metal_matchbox_http_endpoint}/assets/coreos"
    ssh_authorized_key = "${var.tectonic_ssh_authorized_key}"
  }
}

// DO NOT PLACE SECRETS IN USER-DATA

resource "matchbox_group" "controller" {
  count   = "${length(var.tectonic_metal_controller_names)}"
  name    = "${format("%s-%s", var.tectonic_cluster_name, element(var.tectonic_metal_controller_names, count.index))}"
  profile = "${matchbox_profile.tectonic-controller.name}"

  selector {
    mac = "${element(var.tectonic_metal_controller_macs, count.index)}"
    os  = "installed"
  }

  metadata {
    domain_name          = "${element(var.tectonic_metal_controller_domains, count.index)}"
    etcd_name            = "${element(var.tectonic_metal_controller_names, count.index)}"
    etcd_initial_cluster = "${join(",", formatlist("%s=http://%s:2380", var.tectonic_metal_controller_names, var.tectonic_metal_controller_domains))}"
    k8s_dns_service_ip   = "${var.tectonic_kube_dns_service_ip}"
    ssh_authorized_key   = "${var.tectonic_ssh_authorized_key}"

    # extra data
    etcd_image_tag    = "${var.tectonic_versions["etcd"]}"
    kubelet_image_url = "${element(split(":", var.tectonic_container_images["hyperkube"]), 0)}"
    kubelet_image_tag = "${element(split(":", var.tectonic_container_images["hyperkube"]), 1)}"
  }
}

resource "matchbox_group" "worker" {
  count   = "${length(var.tectonic_metal_worker_names)}"
  name    = "${format("%s-%s", var.tectonic_cluster_name, element(var.tectonic_metal_worker_names, count.index))}"
  profile = "${matchbox_profile.tectonic-worker.name}"

  selector {
    mac = "${element(var.tectonic_metal_worker_macs, count.index)}"
    os  = "installed"
  }

  metadata {
    domain_name        = "${element(var.tectonic_metal_worker_domains, count.index)}"
    k8s_dns_service_ip = "${var.tectonic_kube_dns_service_ip}"
    ssh_authorized_key = "${var.tectonic_ssh_authorized_key}"

    # extra data
    kubelet_image_url = "${element(split(":", var.tectonic_container_images["hyperkube"]), 0)}"
    kubelet_image_tag = "${element(split(":", var.tectonic_container_images["hyperkube"]), 1)}"
  }
}
<commit_msg>platforms/metal: Fix missing kube_version_image that was stripped<commit_after>// Install CoreOS to disk
resource "matchbox_group" "coreos-install" {
  count   = "${length(var.tectonic_metal_controller_names) + length(var.tectonic_metal_worker_names)}"
  name    = "${format("coreos-install-%s", element(concat(var.tectonic_metal_controller_names, var.tectonic_metal_worker_names), count.index))}"
  profile = "${matchbox_profile.coreos-install.name}"

  selector {
    mac = "${element(concat(var.tectonic_metal_controller_macs, var.tectonic_metal_worker_macs), count.index)}"
  }

  metadata {
    coreos_channel     = "${var.tectonic_cl_channel}"
    coreos_version     = "${var.tectonic_metal_cl_version}"
    ignition_endpoint  = "${var.tectonic_metal_matchbox_http_endpoint}/ignition"
    baseurl            = "${var.tectonic_metal_matchbox_http_endpoint}/assets/coreos"
    ssh_authorized_key = "${var.tectonic_ssh_authorized_key}"
  }
}

// DO NOT PLACE SECRETS IN USER-DATA

resource "matchbox_group" "controller" {
  count   = "${length(var.tectonic_metal_controller_names)}"
  name    = "${format("%s-%s", var.tectonic_cluster_name, element(var.tectonic_metal_controller_names, count.index))}"
  profile = "${matchbox_profile.tectonic-controller.name}"

  selector {
    mac = "${element(var.tectonic_metal_controller_macs, count.index)}"
    os  = "installed"
  }

  metadata {
    domain_name          = "${element(var.tectonic_metal_controller_domains, count.index)}"
    etcd_name            = "${element(var.tectonic_metal_controller_names, count.index)}"
    etcd_initial_cluster = "${join(",", formatlist("%s=http://%s:2380", var.tectonic_metal_controller_names, var.tectonic_metal_controller_domains))}"
    k8s_dns_service_ip   = "${var.tectonic_kube_dns_service_ip}"
    ssh_authorized_key   = "${var.tectonic_ssh_authorized_key}"

    # extra data
    etcd_image_tag    = "${var.tectonic_versions["etcd"]}"
    kubelet_image_url = "${element(split(":", var.tectonic_container_images["hyperkube"]), 0)}"
    kubelet_image_tag = "${element(split(":", var.tectonic_container_images["hyperkube"]), 1)}"
  }
}

resource "matchbox_group" "worker" {
  count   = "${length(var.tectonic_metal_worker_names)}"
  name    = "${format("%s-%s", var.tectonic_cluster_name, element(var.tectonic_metal_worker_names, count.index))}"
  profile = "${matchbox_profile.tectonic-worker.name}"

  selector {
    mac = "${element(var.tectonic_metal_worker_macs, count.index)}"
    os  = "installed"
  }

  metadata {
    domain_name        = "${element(var.tectonic_metal_worker_domains, count.index)}"
    k8s_dns_service_ip = "${var.tectonic_kube_dns_service_ip}"
    ssh_authorized_key = "${var.tectonic_ssh_authorized_key}"

    # extra data
    kubelet_image_url  = "${element(split(":", var.tectonic_container_images["hyperkube"]), 0)}"
    kubelet_image_tag  = "${element(split(":", var.tectonic_container_images["hyperkube"]), 1)}"
    kube_version_image = "${var.tectonic_container_images["kube_version"]}"
  }
}

{'fraction_non_alphanumeric': 0.1457345971563981, 'fraction_numerical': 0.003385240352064997, 'mean_word_length': 5.22, 'pattern_counts': {'":': 8, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2332186', 'n_tokens_mistral': 2149, 'n_tokens_neox': 2067, 'n_words': 252}